

New locking feature

The new PDFLock, PDFUnlock and PDFsetDocFileProps commands can be used in multi-threading situations to avoid conflicts between multiple application processes or multiple threads requesting simultaneous access to the printer driver. The external library uses the registry to interact with the printer drivers. This can cause conflicts when multiple threads use the external to access the printer drivers. Please read the new PDFWriter reference documentation for a full description about these new commands.

This technical note will show how the printing code has to be changed to use these new commands.

Printing Example

The following is a modified example of the traditional printing code taken from the example libraries that are provided with the software download. The new commands are shown in bold.

The first section of this method has not changed. Things like setting the paper size, orientation, watermark etc can still be set as before and the printer can be selected as before

```
; EXAMPLE: Printing an Omnis report to a PDF file using job locking so
; multiple instances of Omnis can print at the same time
Local variable PrinterName (Character 10000000)
Other parameters are optional
Parameter pShowMessage (Boolean) = kTrue

Working message {Creating PDF file from Omnis report...}

; first set the paper size, etc, if required
PDFsetPaperSize (PDF_PAPER_LETTER)
PDFsetOrientation (PDF_ORIENT_PORTRAIT)
PDFsetWatermark ("Report printed with job locking","Arial",
12,0,rgb(128,128,128),4,4,0)

; now store the settings and select the printer to update Omnis
PDFsetDefaultConfig
PDFgetPrinterName (PrinterName)
Select printer (Discard previous settings) {[PrinterName]}
;; NOTE: you must specify 'Discard previous settings'
```

The next section which traditionally sets the file name and file options have been changed to lock these options for the current process and the new command PDFsetDocFileProps must be used to set them. This command replaces the commands PDFsetDefaultDirectory, PDFsetDefaultFileName and PDFsetFileNameOptions.

```
; Now lock this printjob. We must use the job name as it will appear in
; the print spooler.
PDFLock ("Omnis 7")
```

```
; Once the job is locked we can set the destination file name and
; options we must now use
; PDFsetDocFileProps(Job Title,Options,Directory,Filename)
PDFsetDocFileProps ("Omnis 7","43","","C:\PDFWriterLocking.pdf")
;; Options (43) = PrintWatermark + NoPrompt + UseFileName
```

Printing continues as before.

```
; start the PDF print job ;; this must be called prior to printing
PDFstartJob

; print the report
Send to printer
Set report name rExample
Prepare for print
Print record
Print record
End print
```

Once the report has been printed we must now call the external to unlock the file options, in case they have not been unlocked already by the printer driver due to an error.

```
; unlock the print job
; we specify a timeout long enough to allow the job to complete
PDFUnlock ("Omnis 7",1000)
```

Finally we finish the print job just as before and display a final message

```
;; PDFendJob must be called after printing is complete
PDFendJob

; close working message and display a final message
Close working message
If pShowMessage
    OK message (High position, Large size, Sound bell) {To view the PDF
    please double click the file C:\PDFWriterLocking.pdf on your disk. You
    will require a PDF reader to view this file.}
End If
```

Document History

21 February 2019: first publication